NAG Library Routine Document

C06FCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C06FCF calculates the discrete Fourier transform of a sequence of n complex data values (using a work array for extra speed).

2 Specification

```
SUBROUTINE C06FCF (X, Y, N, WORK, IFAIL)
INTEGER                 N, IFAIL
REAL (KIND=nag_wp) X(N), Y(N), WORK(N)
```

3 Description

Given a sequence of n complex data values z_j, for $j = 0, 1, \ldots, n-1$, C06FCF calculates their discrete Fourier transform defined by

$$\hat{z}_k = a_k + ib_k = \frac{1}{\sqrt{n}} \sum_{j=0}^{n-1} z_j \times \exp\left(-i\frac{2\pi jk}{n}\right), \quad k = 0, 1, \ldots, n-1.$$

(Note the scale factor of $\frac{1}{\sqrt{n}}$ in this definition.)

To compute the inverse discrete Fourier transform defined by

$$\hat{w}_k = \frac{1}{\sqrt{n}} \sum_{j=0}^{n-1} z_j \times \exp\left(+i\frac{2\pi jk}{n}\right),$$

this routine should be preceded and followed by the complex conjugation of the data values and the transform (by negating the imaginary parts stored in y).

C06FCF uses the fast Fourier transform (FFT) algorithm (see Brigham (1974)). There are some restrictions on the value of n (see Section 5).

4 References

Brigham E O (1974) *The Fast Fourier Transform* Prentice–Hall

5 Arguments

1: X(N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: if X is declared with bounds $(0 : \mathrm{N}-1)$ in the subroutine from which C06FCF is called, then $\mathrm{X}(j)$ must contain x_j, the real part of z_j, for $j = 0, 1, \ldots, n-1$.

On exit: the real parts a_k of the components of the discrete Fourier transform. If X is declared with bounds $(0 : \mathrm{N}-1)$ in the subroutine from which C06FCF is called, then for $0 \leq k \leq n-1$, a_k is contained in $\mathrm{X}(k)$.

2: Y(N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: if Y is declared with bounds $(0 : \mathrm{N}-1)$ in the subroutine from which C06FCF is called, then $\mathrm{Y}(j)$ must contain y_j, the imaginary part of z_j, for $j = 0, 1, \ldots, n-1$.

On exit: the imaginary parts b_k of the components of the discrete Fourier transform. If Y is declared with bounds $(0 : \mathrm{N} - 1)$ in the subroutine from which C06FCF is called, then for $0 \leq k \leq n - 1$, b_k is contained in $\mathrm{Y}(k)$.

3: N – INTEGER *Input*

On entry: n, the number of data values. The largest prime factor of N must not exceed 19, and the total number of prime factors of N, counting repetitions, must not exceed 20.

Constraint: $\mathrm{N} > 1$.

4: WORK(N) – REAL (KIND=nag_wp) array *Workspace*

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\mathrm{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\mathrm{IFAIL} = 1$

At least one of the prime factors of N is greater than 19.

$\mathrm{IFAIL} = 2$

N has more than 20 prime factors.

$\mathrm{IFAIL} = 3$

On entry, $\mathrm{N} \leq 1$.

$\mathrm{IFAIL} = 4$

An unexpected error has occurred in an internal call. Check all subroutine calls and array dimensions. Seek expert help.

$\mathrm{IFAIL} = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$\mathrm{IFAIL} = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Some indication of accuracy can be obtained by performing a subsequent inverse transform and comparing the results with the original sequence (in exact arithmetic they would be identical).

8 Parallelism and Performance

C06FCF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

C06FCF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The time taken is approximately proportional to $n \times \log(n)$, but also depends on the factorization of n. C06FCF is faster if the only prime factors of n are 2, 3 or 5; and fastest of all if n is a power of 2.

10 Example

This example reads in a sequence of complex data values and prints their discrete Fourier transform (as computed by C06FCF). It then performs an inverse transform using C06FCF, and prints the sequence so obtained alongside the original data values.

10.1 Program Text

```
    Program c06fcfe

!     C06FCF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: c06fcf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: ieof, ifail, j, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: work(:), x(:), xx(:), y(:), yy(:)
!     .. Executable Statements ..
      Write (nout,*) 'C06FCF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
loop: Do
        Read (nin,*,Iostat=ieof) n
        If (ieof<0) Then
          Exit loop
        End If

        Allocate (x(0:n-1),xx(0:n-1),y(0:n-1),yy(0:n-1),work(n))
```

```
        Read (nin,*)(x(j),y(j),j=0,n-1)
        xx(0:n-1) = x(0:n-1)
        yy(0:n-1) = y(0:n-1)

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call c06fcf(x,y,n,work,ifail)

        Write (nout,*)
        Write (nout,*) 'Components of discrete Fourier transform'
        Write (nout,*)
        Write (nout,*) '          Real      Imag'
        Write (nout,*)
        Write (nout,99999)(j,x(j),y(j),j=0,n-1)

        y = -y
        Call c06fcf(x,y,n,work,ifail)
        y = -y

        Write (nout,*)
        Write (nout,*) 'Original sequence as restored by inverse transform'
        Write (nout,*)
        Write (nout,*) '             Original                 Restored'
        Write (nout,*) '          Real      Imag           Real      Imag'
        Write (nout,*)
        Write (nout,99998)(j,xx(j),yy(j),x(j),y(j),j=0,n-1)
        Deallocate (x,xx,y,yy,work)
      End Do loop

99999 Format (1X,I5,2F10.5)
99998 Format (1X,I5,2F10.5,5X,2F10.5)
    End Program c06fcfe
```

10.2 Program Data

```
C06FCF Example Program Data
    7                 : n
  0.34907  -0.37168
  0.54890  -0.35669
  0.74776  -0.31175
  0.94459  -0.23702
  1.13850  -0.13274
  1.32850   0.00074
  1.51370   0.16298    : x, y
```

10.3 Program Results

```
 C06FCF Example Program Results

 Components of discrete Fourier transform

           Real      Imag

     0   2.48361  -0.47100
     1  -0.55180   0.49684
     2  -0.36711   0.09756
     3  -0.28767  -0.05865
     4  -0.22506  -0.17477
     5  -0.14825  -0.30840
     6   0.01983  -0.56496

 Original sequence as restored by inverse transform

              Original                Restored
           Real      Imag           Real      Imag

     0   0.34907  -0.37168        0.34907  -0.37168
     1   0.54890  -0.35669        0.54890  -0.35669
```

```
     2   0.74776  -0.31175        0.74776  -0.31175
     3   0.94459  -0.23702        0.94459  -0.23702
     4   1.13850  -0.13274        1.13850  -0.13274
     5   1.32850   0.00074        1.32850   0.00074
     6   1.51370   0.16298        1.51370   0.16298
```